
One of the new features provided by Clipper 5.01's preprocessor is conditional compilation, a feature borrowed directly from the 

C language which allows you to determine which portions of source code in a given file to compile. Conditional compilation 

allows you to maintain more than one version of a program in a single source file.  

 

If you're like most Clipper developers who create and maintain their own function libraries, you could use conditional 

compilation to test code contained in each function's source file. This method of maintaining test code is useful for modifying 

the function, for demonstrating it, or simply determining how to call it. 

 

Because the test code is developed when the function is written, you can easily take into consideration any assumptions, and 
test for the boundary conditions of each function. After a function is tested and ready for inclusion in a library, it can be 

compiled into an object module devoid of any testing routines. 

 

The following source code provides a simple example of this technique using a function called AInvert() stored in the file  

 
AINVERT.PRG: 
 
/* AINVERT.PRG - Returns an inverted array, one level only.  
 
SYNTAX : 
 
    AInvert( ) -> NIL 
 
COMPILING : 
 
   FOR LIBRARY: /n /w /m 
   FOR TESTING: /n /w /m /dTESTING 
 
*/ 
 
#ifdef TESTING 
 
   PROCEDURE Test    /* Should output : Red Blue Green Green Blue Red */ 
 
      LOCAL aArray := { "Red", "Green", "Blue" } 
      LOCAL aInvert := AInvert ( aArray ) 
      LOCAL I 
 
      FOR i := 1 TO Len ( aArray ) 
        ? aArray [ i ], aInvert [ i ] 
      NEXT 
 
   RETURN  
 
#endif 
 
FUNCTION AInvert ( aArray ) 
 
   LOCAL n := Len ( aArray ) 
   LOCAL aInverse[ n ] 
 
   AEval ( aArray, { | e | aInverse [ n-- ] := e } ) 
 
RETURN aInverse 
 
// EOF 
 
Note that the conditional compilation is provided via the #ifdef and #endif directives and the manifest constant TESTING. To 

compile, link, and run AInvert() with the test routines, simply use the /d option and specify TESTING when you compile: 

 
CLIPPER ainvert /n /w /m /dTESTING RTLINK FILE ainvert ainvert 
 

Once AInvert() has been fully tested stand-alone, simply recompile without the /dTESTING,: 

 
CLIPPER ainvert /n /w /m 
 

and your function sans test code is ready for inclusion in your library. 

